


## Способы нахождения неизвестных значений посредством интерполяции в Excel



По роду своей деятельности, большинство специалистов, задействованных в области инженерных изысканий, и инженеры-геологи в частности, сталкиваются с большим количеством данных. Такие данные, иногда, для удобства располагаются в таблицах, которые можно наблюдать, к примеру, в нормативной документации. Но порой, массива данных бывает недостаточно для получения необходимых значений и здесь на помощь, в некоторых случаях, может прийти такой математический инструмент, как интерполяция.

В процессе взаимодействия с некоторыми сторонними организациями, выполняющими инженерно-геологические и другие виды изысканий, отмечено, что большое количество расчетов интерполяции производится в ручном режиме. В лучшем случае, на листе бумаги, в худшем - что называется, «на глаз». И то и другое не очень хорошо. Эти рутинные действия отнимают большое количество времени, сил, а так же не исключают допущения ошибок в полученном результате.

Поэтому возникла идея подготовки и публикации данного материала. В нем будет рассмотрена одна из возможностей использования программы Excel для нахождения промежуточных значений, которая входит в стандартный пакет MS Office, установленный в большинстве компаний.

Вы научитесь делать это самостоятельно, создадите свой файл с расчетами, оцените все его преимущества, и сведете к минимуму, так называемый, человеческий фактор. Сам материал будет изложен общедоступным языком, а формулы - упрощены для лучшего их понимания.

Все примеры приводятся в русскоязычной версии MS Office Excel 2013, однако и в более ранних, большинство формул должно работать, возможно за исключением тех, в которых будет задано большое количество условий.

Заместитель начальника изыскательского отдела по геологии ППОНИГМ  
ООО «ОренбургНИПИнефть», г. Оренбург, Россия, [arapovgeo@gmail.com](mailto:arapovgeo@gmail.com)

## Введение


Интерполяция – в вычислительной математике способ нахождения промежуточных значений величины по имеющемуся дискретному набору известных значений. И в университете, на парах математики вы с ней скорее всего познакомились.

Вариантов интерполяции существует некоторое множество, но мы будем рассматривать именно линейную интерполяцию, которую в Excel можно выполнять с помощью функции ПРЕДСКАЗ. Стоит отметить, что сама эта функция имеет более расширенные возможности.

Для наглядности построим простую точечную диаграмму с прямыми отрезками и маркерами. А значения для нее возьмем из таблицы 5.1, СП 22.13330.2016 «Основания зданий и сооружений» Актуализированная редакция СНиП 2.02.01-83\* (с изменениями №1, 2, 3). 2019 г. Возможно, это один из самых частых примеров использования интерполяции, с помощью которой получают значение коэффициента  $m_{oed}$ , применяемого для корректировки одометрического модуля деформации.

Сначала мы будем использовать не все данные из этой таблицы, а только часть. Например – значения  $m_{oed}$  для супесей при коэффициенте пористости  $e$  0,65-0,75. Создайте такую же таблицу в Excel. Обратите внимание, что содержимое должно соответствовать тем же строкам и столбцам, что и в примере.

На оси X в данном случае будут располагаться значения коэффициента пористости  $e$ , а на оси Y – коэффициента  $m_{oed}$ , соответственно. Посаженные по координатам точки будут соединены отрезком, который мы условно обозначим  $ab$  (рис.1).



**Рис. 1.** Точечная диаграмма по двум значениям

Давайте представим, что нам необходимо найти  $m_{oed}$  для супеси с коэффициентом пористости 0,7. Для этого, от числа 0,7 на оси X мы проведем параллельную оси Y линию  $fc$  до нашего отрезка. Затем, от точки пересечения, проведем к оси Y уже параллельно оси X линию  $cd$ . И получим значение  $m_{oed}$  – 2,3 графика (рис. 2).




Рис. 2. Пример графического метода интерполяции

Это графический способ. Математически, формула линейной интерполяции, в данном случае, выглядит так:

$$Y = Y_1 + \frac{(X - X_1)}{(X_0 - X_1)} * (Y_0 - Y_1)$$

где  $Y_0=2,1$ ;  $Y_1=2,5$ ;  $X=0,7$ ;  $X_0=0,75$ ;  $X_1=0,65$

На рисунке 3, приведена диаграмма с соответствующими обозначениями.




Рис. 3. Точечная диаграмма с обозначениями для примера математического метода интерполяции

Подставив все эти значения в формулу, получаем:

$$Y = 2,5 + \frac{(0,7 - 0,65)}{(0,75 - 0,65)} * (2,1 - 2,5) = 2,3$$

$Y$  - есть наше искомое значение *m<sub>oed</sub>* для коэффициента пористости 0,70, которое для супеси равно 2,3.

Удобен ли такой расчет на листе бумаги? Не очень, т.к. отнимает много времени. Однако уже хорошо, что он вообще выполняется.


На данный момент, когда уже понятно, как выглядит линейная интерполяция графически и как она рассчитывается математически, для упрощения используем функцию ПРЕДСКАЗ, которая может сделать то же самое для двух наших значений  $e$  и соответствующих им  $m_{oed}$ . Сама функция в Excel имеет следующий вид:

**ПРЕДСКАЗ(х;известные\_значения\_у;известные\_значения\_x)**

Возвращаемся к нашему примеру и в ячейку M5 запишем известное значение коэффициента пористости – 0,70, а в ячейку N5 впишем следующую формулу:

**=ПРЕДСКАЗ(M5;M3:N3;M2:N2)**

Нажмем Enter. В результате, в ячейке N5 получим значение 2,3, которое соответствует нашему искомому коэффициенту  $m_{oed}$ . Изменяя значение в ячейке M5 от 0,65 до 0,75, вы будете получать в соответствии с ним новые значения в ячейке N5 (рис. 4).



**Рис. 4.** Окончательный результат интерполяции с помощью функции ПРЕДСКАЗ

Теперь полностью автоматизируем расчеты и разберем на примерах различные варианты. Для каждого из них создадим отдельный лист в книге Excel.

**Важно! Все значения должны быть размещены в тех же строках и столбцах, что и в примерах.**


### Пример 1. Получение коэффициента $m_{oed}$

Построим таблицу 5.1 из пункта 5.3.7, СП 22.13330.2016 «Основания зданий и сооружений» Актуализированная редакция СНиП 2.02.01-83\* (с изменениями №1, 2, 3). 2019 г. и автоматизируем получение коэффициента  $m_{oed}$  в зависимости от коэффициента пористости  $e$  и выбранного типа грунта. Для удобства, столбец «0,45-0,55» разобьем на два. Там, где значений нет, поставим 0 (рис. 5).


	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1							Значение коэффициента $m_{oed}$ при коэффициенте пористости $e$ , равном							
2						Грунты	0,45	0,55	0,65	0,75	0,85	0,95	1,05	
3														
4						Супеси	2,8	2,8	2,5	2,1	1,4	0,0	0,0	
5						Суглинки	3,0	3,0	2,7	2,4	2,0	1,5	1,0	
6						Глины	0,0	0,0	2,4	2,3	2,2	2,0	1,8	
7														

**Рис. 5.** Таблица зависимости  $m_{oed}$  от коэффициента пористости для некоторых грунтов


Так, как зависимость значений в таблице не линейная и это наглядно видно, если построить по ним все ту же диаграмму, и выполнить линейную аппроксимацию (рис. 6, 7, 8), мы не можем взять сразу весь массив данных.



**Рис. 6.** Точечная диаграмма и линейная аппроксимация по значениям из таблицы 5.1, СП 22.13330.2016 для супесей



**Рис. 7.** Точечная диаграмма и линейная аппроксимация по значениям из таблицы 5.1, СП 22.13330.2016 для суглинков



**Рис. 8.** Точечная диаграмма и линейная аппроксимация по значениям из таблицы 5.1, СП 22.13330.2016 для глин

Поэтому, сначала выберем по два значения, внутри диапазона которых будет проведена интерполяция. А затем сделаем это для всех данных из таблицы, по очереди.

Далее будет указано, в какую ячейку, какое содержимое вписывать. Формат ячеек может быть, как общий, так и числовой.

Ячейка	Содержимое ячейки
A2	Выберите грунт
A3	Коэфф. пористости
B3	0,75
A6	Коэфф. moed
H8	=ПРЕДСКАЗ(B3;H4:I4;H3:I3)
I8	=ПРЕДСКАЗ(B3;I4:J4;I3:J3)
J8	=ПРЕДСКАЗ(B3;J4:K4;J3:K3)
H9	=ПРЕДСКАЗ(B3;H5:I5;H3:I3)
I9	=ПРЕДСКАЗ(B3;I5:J5;I3:J3)
J9	=ПРЕДСКАЗ(B3;J5:K5;J3:K3)
K9	=ПРЕДСКАЗ(B3;K5:L5;K3:L3)
L9	=ПРЕДСКАЗ(B3;L5:M5;L3:M3)
I10	=ПРЕДСКАЗ(B3;I6:J6;I3:J3)
J10	=ПРЕДСКАЗ(B3;J6:K6;J3:K3)
K10	=ПРЕДСКАЗ(B3;K6:L6;K3:L3)
L10	=ПРЕДСКАЗ(B3;L6:M6;L3:M3)
F8	=ЕСЛИ(B3<=0,55;G4;ЕСЛИ(И(B3>0,55;B3<=0,65);H8;ЕСЛИ(И(B3>0,65;B3<=0,75);I8;ЕСЛИ(И(B3>0,75;B3<=0,85);J8;ЕСЛИ(B3>0,85;"Ошибка")))))
F9	=ЕСЛИ(B3<=0,55;H5;ЕСЛИ(И(B3>0,55;B3<=0,65);H9;ЕСЛИ(И(B3>0,65;B3<=0,75);I9;ЕСЛИ(И(B3>0,75;B3<=0,85);J9;ЕСЛИ(И(B3>0,85;B3<=0,95);K9;ЕСЛИ(И(B3>0,95;B3<=1,05);L9))))))
F10	=ЕСЛИ(B3<0,65;"Ошибка";ЕСЛИ(И(B3>=0,65;B3<=0,75);I10;ЕСЛИ(И(B3>0,75;B3<=0,85);J10;ЕСЛИ(И(B3>0,85;B3<=0,95);K10;ЕСЛИ(И(B3>0,95;B3<=1,05);L10))))))

Затем, для большего удобства, сделаем выбор типа грунта из выпадающего списка. Для этого:

- 1) Выберите ячейку B2
- 2) На вкладке «Данные», в разделе «Работа с данными», выберите «Проверка данных» (рис. 9)




Рис. 9. Расположение пункта «Проверка данных» в Excel 2013

- 3) В появившемся окне, в разделе «Условия проверки», в качестве типа данных установите «Список»
- 4) В строку «Источник» впишите:  
 $=\$F\$4:\$F\$6$
- 5) Нажмите «Ок» (рис. 10)




Рис. 10. Меню «Проверка вводимых значений» в Excel 2013

Остается только получить нужный результат. Впишите формулу в ячейку В6:

$=\text{ЕСЛИ}(\text{B2}=\text{"Супеси"};\text{F8};\text{ЕСЛИ}(\text{B2}=\text{"Суглинки"};\text{F9};\text{ЕСЛИ}(\text{B2}=\text{"Глины"};\text{F10})))$

Если вы все сделали правильно, то ваш лист Excel должен иметь следующий вид (рис. 11):

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1														
2	Выберите грунт	Глины				Грунты	Значение коэффициента $m_{oed}$ при коэффициенте пористости $e$ , равном							
3	Коэфф. пористости	0,75					0,45	0,55	0,65	0,75	0,85	0,95	1,05	
4						Супеси	2,8	2,8	2,5	2,1	1,4	0,0	0,0	
5	Результат					Суглинки	3,0	3,0	2,7	2,4	2,0	1,5	1,0	
6	Коэфф. $m_{oed}$	2,3				Глины	0,0	0,0	2,4	2,3	2,2	2,0	1,8	
7														
8						2,10		2,2	2,1	2,1				
9						2,4		2,4	2,4	2,5	2,5			
10						2,3			2,3	2,4	2,4			
11														

Рис. 11. Конечный вид примера 1 на листе Excel

Теперь, выбирая грунт из списка в ячейке В2 и указав нужный коэффициент пористости в ячейке В3, вы получите искомое значение коэффициента  $m_{oed}$  в ячейке В6. В случае отсутствия значений, будет выводиться сообщение об ошибке.

Стоит отметить, что добиться подобного можно и другими способами. Например, используя связку функций Excel: ГПР, ИНДЕКС и ПОИСКПОЗ. При этом не потребуется интерполировать всю таблицу, а условных операторов будет меньше. Рассмотрим этот вариант ниже.

**Пример 2. Получение расчетного сопротивления глинистых непросадочных грунтов**  
Построим таблицу Б3 из приложения Б, СП 22.13330.2016 «Основания зданий и сооружений» Актуализированная редакция СНиП 2.02.01-83\* (с изменениями №1, 2, 3). 2019 г. и автоматизируем получение  $R_0$  в зависимости от коэффициента пористости  $e$ , показателя текучести  $I_L$  и выбранного типа грунта (рис. 12). Для удобства, в ячейках Н2 и И2 оставим только цифры.



	A	B	C	D	E	F	G	H		I
						Глинистые грунты	Коэффициент пористости $e$	Значение $R_0$ , кПа при показателе текучести грунта $I_L$		
						0		1		
1										
2										
3						Супеси	0,5	300	200	
4							0,7	250	150	
5						Суглинки	0,5	350	250	
6							0,7	250	180	
7							1	200	100	
8						Глины	0,5	600	400	
9							0,6	500	300	
10							0,8	300	200	
11							1,1	250	100	

Рис. 12. Таблица Б3 из СП 22.13330.2016

Далее будет указано, в какую ячейку, какое содержимое вписывать. Формат ячеек может быть, как общий, так и числовой.

Ячейка	Содержимое ячейки
A2	Выберите грунт
A3	$I_L$
A4	$e$
A5	$R_0$ , кПа
B3	0,6
B4	0,5
G13	=ПРЕДСКАЗ(B4;H3:H4;G3:G4)
H13	=ПРЕДСКАЗ(B4;I3:I4;G3:G4)
J13	=ПРЕДСКАЗ(B3;G13:H13;H2:I2)
J14	=ЕСЛИ(ИЛИ(B3>1;B4>0,7;B4<0,5);"Ошибка";J13)
F15	=ВПР(B4;G5:G7;1)
G15	=ИНДЕКС(\$H\$5:\$I\$7;ПОИСКПОЗ(\$F15;\$G\$5:\$G\$7);ПОИСКПОЗ(H\$2;\$H\$2:\$I\$2;))
H15	=ИНДЕКС(\$H\$5:\$I\$7;ПОИСКПОЗ(\$F15;\$G\$5:\$G\$7);ПОИСКПОЗ(I\$2;\$H\$2:\$I\$2;))
F16	=ИНДЕКС(G5:G7;ПОИСКПОЗ(1;ЧАСТОТА(B4;G5:G7;)))
G16	=ИНДЕКС(\$H\$5:\$I\$7;ПОИСКПОЗ(\$F16;\$G\$5:\$G\$7);ПОИСКПОЗ(H\$2;\$H\$2:\$I\$2;))
H16	=ИНДЕКС(\$H\$5:\$I\$7;ПОИСКПОЗ(\$F16;\$G\$5:\$G\$7);ПОИСКПОЗ(I\$2;\$H\$2:\$I\$2;))
G17	=ПРЕДСКАЗ(B4;G15:G16;\$F15:\$F16)
H17	=ПРЕДСКАЗ(B4;H15:H16;\$F15:\$F16)
J15	=ПРЕДСКАЗ(B3;G17:H17;H2:I2)
J16	=ЕСЛИ(И(H2<>I2;F15=F16);J15;ЕСЛИ(И(H2<>I2;F15<>F16);J16))
J17	=ЕСЛИ(ИЛИ(B3>1;B4>1;B4<0,5);"Ошибка";J17)
F20	=ВПР(B4;G8:G11;1)
G20	=ИНДЕКС(\$H\$8:\$I\$11;ПОИСКПОЗ(\$F20;\$G\$8:\$G\$11);ПОИСКПОЗ(H\$2;\$H\$2:\$I\$2;))
H20	=ИНДЕКС(\$H\$8:\$I\$11;ПОИСКПОЗ(\$F20;\$G\$8:\$G\$11);ПОИСКПОЗ(I\$2;\$H\$2:\$I\$2;))



F21	=ИНДЕКС(G8:G11;ПОИСКПОЗ(1;ЧАСТОТА(B4;G8:G11);))
G21	=ИНДЕКС(\$H\$8:\$I\$11;ПОИСКПОЗ(\$F21;\$G\$8:\$G\$11);ПОИСКПОЗ(H\$2;\$H\$2:\$I\$2;))
H21	=ИНДЕКС(\$H\$8:\$I\$11;ПОИСКПОЗ(\$F21;\$G\$8:\$G\$11);ПОИСКПОЗ(I\$2;\$H\$2:\$I\$2;))
G22	=ПРЕДСКАЗ(B4;G20:G21;\$F20:\$F21)
H22	=ПРЕДСКАЗ(B4;H20:H21;\$F20:\$F21)
J20	=ПРЕДСКАЗ(B3;G20:H20;H2:I2)
J21	=ПРЕДСКАЗ(B3;G22:H22;H2:I2)
J22	=ЕСЛИ(И(H2<>I2;F20=F21);J20;ЕСЛИ(И(H2<>I2;F20<>F21);J21))
J23	=ЕСЛИ(ИЛИ(B3>1;B4>1,1;B4<0,5);"Ошибка";J22)

Теперь выберем ячейку B2 и создадим выпадающий список грунтов, как в первом примере. Только в поле «Источник» (см. рис. 10) впишем уже такое содержимое:

Супеси;Суглинки;Глины

Подтвердите все кнопкой «Ок» и нажмите на ячейку B2. Появится выпадающий список, в котором выберите, например, суглинки.

Остается только получить  $R_0$ . Для этого, впишите такую формулу в ячейку B5:

=ЕСЛИ(B2="Супеси";J14;ЕСЛИ(B2="Суглинки";J18;ЕСЛИ(B2="Глины";J23)))

В результате, лист Excel должен иметь следующий вид (рис. 13).

	A	B	C	D	E	F	G	H		I	J
						Глинистые грунты	Коэффициент пористости e	Значение R0, кПа при показателе текучести грунта IL			
								0	1		
1											
2	Выберите грунт	Суглинки									
3	IL		0,6			Супеси	0,5	300	200		
4	e		0,5				0,7	250	150		
5	R0, кПа		290,0			Суглинки	0,5	350	250		
6							0,7	250	180		
7							1	200	100		
8						Глины	0,5	600	400		
9							0,6	500	300		
10							0,8	300	200		
11							1,1	250	100		
12											
13								300	200		240
14											240
15							0,5	350	250		290
16							0,5	350	250		#ДЕЛ/0!
17								#ДЕЛ/0!	#ДЕЛ/0!		290
18											290
19											
20							0,5	600	400		480
21							0,5	600	400		#ДЕЛ/0!
22								#ДЕЛ/0!	#ДЕЛ/0!		480
23											480

Рис. 13. Конечный вид примера 2 на листе Excel

Как можно заметить, в этом примере, для того чтобы не выполнять поочередную интерполяцию всех значений таблицы, были использованы функции ВПР, ИНДЕКС и ПОИСКПОЗ, что значительно упростило задачу. Изменяя тип грунта, а так же значения в ячейках B3 и B4, вы неизменно получите значение расчетного сопротивления в ячейке B5.

### Пример 3. Косвенный метод определения плотности $\rho$ песков по результатам статического зондирования

Широко известно, что отобрать образцы песка действительно ненарушенной структуры из инженерно-геологических скважин, даже на сегодняшний день, чрезвычайно

затруднительно, особенно на значительных глубинах и в случаях, когда такие грунты обводнены. Это подтверждал и Лев Геннадьевич Мариупольский в книге «Исследования грунтов для проектирования и строительства свайных фундаментов», Москва, Стройиздат 1989.

В качестве альтернативы лабораторному методу определения плотности природного сложения аллювиальных и флювиогляциальных песков, залегающих на глубине до 6 м, Л.Г. Мариупольским была предложена возможность определения  $p$  с помощью результатов статического зондирования, природной влажности ( $W$ ) и плотности частиц грунта ( $p_s$ ). Сопоставив 171 определение коэффициента пористости  $e$ , полученного в лабораторных условиях из ненарушенных образцов проб песка с сопротивлением конусу ( $q_c$ ) зонда II типа этих же грунтов, выведена следующая корреляционная зависимость:

$$e = 0.765 - 0.185 \lg \left( \frac{q_c}{1} \right)$$

где 1 – это 1 МПа.

По утверждениям автора, коэффициент корреляции для такой зависимости составил 0,74, а среднее квадратическое отклонение – 0,09. Причем, гранулометрический состав, в таком случае, практически не влияет на точность определения  $e$ .

Исходя из вышесказанного, получена формула количественного определения плотности  $p$  таких песков:

$$p = \frac{p_s(1 + W)}{1 + e}$$

$p_s$ - плотность частиц грунта;

$W$ - влажность грунта;

$e$  – коэффициент пористости.

Если подставить в эту формулу коэффициент пористости, то получится следующее:

$$p = \frac{p_s(1 + W)}{1.765 - 0.185 \lg \left( \frac{q_c}{1} \right)}$$

Т.к. для определения  $W$  и  $p_s$  нет необходимости отбирать образцы ненарушенной структуры, то, по утверждениям автора, точности результатов  $p$  по данной формуле достаточно для расчетов оснований свайных фундаментов.

В следствии этого была представлена таблица, в которой приведены значения плотности  $p$  песков в зависимости от их  $q_c$  и  $W$ .

Таблица 1

**Значения плотности песков  $p$  в зависимости от сопротивления конусу зонда  $q_c$  и влажности  $W$  по Мариупольскому**

$W$	$p$ , г/см <sup>3</sup> , при сопротивлении $q_c$ , МПа									
	1	2	3	5	10	15	20	30	40	50
0	1,5	1,55	1,57	1,61	1,68	1,72	1,73	1,78	1,8	1,83
0,05	1,58	1,63	1,65	1,7	1,76	1,8	1,82	1,87	1,89	1,92
0,1	1,65	1,71	1,74	1,78	1,84	1,88	1,91	1,95	1,98	2,0
0,15	1,72	1,78	1,82	1,86	1,93	1,97	1,99	2,01	2,07	2,1
0,2	1,8	1,86	1,89	1,94	2,01	2,05	2,08	-	-	-
0,25	1,88	1,94	1,96	1,94	-	-	-	-	-	-
0,3	1,95	1,95	1,96	1,96	-	-	-	-	-	-

Для расчета значений в этой таблице, плотность частиц песков принималась, как средняя,  $p_s=2,65$  г/см<sup>3</sup>.

Это было небольшое отступление и в данном примере не рассматривается вариант применения такого расчета для определения плотности песков. Цель – показать возможность интерполяции по двум неизвестным, для которого хорошо подходит такая таблица.

Теперь перенесем ее на новый лист Excel. Чтобы еще немного усложнить задачу, расположим значения в строке  $q_c$  от большего, к меньшему. Соответственно, значения  $p$  тоже перенесем (рис. 14).

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1				w	p, г/см3, при сопротивлении qc, МПа									
2					50	40	30	20	15	10	5	3	2	1
3				0	1,83	1,8	1,78	1,73	1,72	1,68	1,61	1,57	1,55	1,5
4				0,05	1,92	1,89	1,87	1,82	1,8	1,76	1,7	1,65	1,63	1,58
5				0,1	2	1,98	1,95	1,91	1,88	1,84	1,78	1,74	1,71	1,65
6				0,15	2,1	2,07	2,01	1,99	1,97	1,93	1,86	1,82	1,78	1,72
7				0,2				2,08	2,05	2,01	1,94	1,89	1,86	1,8
8				0,25							1,94	1,96	1,94	1,88
9				0,3							1,96	1,96	1,95	1,95

Рис. 14. Таблица 1 на листе Excel

Далее ход действий такой же, как и в предыдущих примерах. Формат ячеек может быть, как общий, так и числовой.

Ячейка	Содержимое ячейки
A1	Влажность природ. W, д.е.
A2	qc (по статике), МПа
A3	Плотность частиц, г/см3
B1	0,05
B2	20
B3	2,65
F13	=ВПР(B1;D3:D9;1)
F14	=ИНДЕКС(D3:D9;ПОИСКПОЗ(1;ЧАСТОТА(B1;D3:D9);))
G12	=ИНДЕКС(E2:N2;СЧЁТЕСЛИ(E2:N2;">"&B2)+1)
G13	=ИНДЕКС(\$E\$3:\$N\$9;ПОИСКПОЗ(\$F13;\$D\$3:\$D\$9;);ПОИСКПОЗ(G\$12;\$E\$2:\$N\$2;))
G14	=ИНДЕКС(\$E\$3:\$N\$9;ПОИСКПОЗ(\$F14;\$D\$3:\$D\$9;);ПОИСКПОЗ(G\$12;\$E\$2:\$N\$2;))
G15	=ПРЕДСКАЗ(\$B1;G13:G14;\$F13:\$F14)
H12	=ИНДЕКС(E2:N2;ПОИСКПОЗ(B2;E2:N2;-1))
H13	=ИНДЕКС(\$E\$3:\$N\$9;ПОИСКПОЗ(\$F13;\$D\$3:\$D\$9;);ПОИСКПОЗ(H\$12;\$E\$2:\$N\$2;))
H14	=ИНДЕКС(\$E\$3:\$N\$9;ПОИСКПОЗ(\$F14;\$D\$3:\$D\$9;);ПОИСКПОЗ(H\$12;\$E\$2:\$N\$2;))
H15	=ПРЕДСКАЗ(\$B1;H13:H14;\$F13:\$F14)
K12	=ПРЕДСКАЗ(B2;G13:H13;G12:H12)
K14	=ПРЕДСКАЗ(\$B1;G13:G14;\$F13:\$F14)
K16	=G13
K18	=ПРЕДСКАЗ(B2;G15:H15;G12:H12)

Теперь обрабатываем результаты, а заодно рассчитываем такие показатели, как плотность сухого грунта  $\rho_d$ , коэффициент пористости  $e$  и коэффициент водонасыщения  $S_r$ .

Ячейка	Содержимое ячейки
A5	Результат
A7	Плотность природ., г/см <sup>3</sup>
A8	Плотность сухого, г/см <sup>3</sup>
A9	Коэфф. пористости
A10	Коэфф. водонасыщ.
B7	=ЕСЛИ(И(G12<>H12;F13=F14);K12;ЕСЛИ(И(F13<>F14;G12=H12);K14;ЕСЛИ(И(F13=F14;G12=H12);K16;ЕСЛИ(И(G12<>H12;F13<>F14);K18))))
B8	=B7/(1+B1)
B9	=(B3-B8)/B8
B10	=(B1*B3)/B9

Дополнительно, помимо простой интерполяции, выполним расчет  $\rho$  по тем формулам, которые указывались выше:

Ячейка	Содержимое ячейки
A13	Расчет плотности по Мариупольскому
A15	Коэфф. пористости
A16	Плотность природ., г/см <sup>3</sup>
A17	Плотность сухого, г/см <sup>3</sup>
A18	Коэфф. водонасыщ
B15	=0,765-0,185*LOG(B2/1)
B16	=(B3*(1+B1))/(1,765-0,185*LOG(B2/1))
B17	=B16/(1+B1)
B18	=(B1*B3)/B15

Теперь, лист Excel у вас должен иметь следующий вид (рис. 15)

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	Влажность природ. W, д.е.	0,050		w	ρ, г/см <sup>3</sup> , при сопротивлении qc, МПа									
2	qc (по статике), МПа	20			50	40	30	20	15	10	5	3	2	1
3	Плотность частиц, г/см <sup>3</sup>	2,65		0	1,83	1,8	1,78	1,73	1,72	1,68	1,61	1,57	1,55	1,5
4				0,05	1,92	1,89	1,87	1,82	1,8	1,76	1,7	1,65	1,63	1,58
5	Результат интерполяции			0,1	2	1,98	1,95	1,91	1,88	1,84	1,78	1,74	1,71	1,65
6				0,15	2,1	2,07	2,01	1,99	1,97	1,93	1,86	1,82	1,78	1,72
7	Плотность природ., г/см <sup>3</sup>	1,82		0,2				2,08	2,05	2,01	1,94	1,89	1,86	1,8
8	Плотность сухого, г/см <sup>3</sup>	1,73		0,25							1,94	1,96	1,94	1,88
9	Коэфф. пористости	0,53		0,3							1,96	1,96	1,95	1,95
10	Коэфф. водонасыщ.	0,25												
11														
12							20				#ДЕЛ/0!			
13	Расчет плотности по Мариупольскому				0,05	1,82	1,82							
14					0,05	1,82	1,82							
15	Коэфф. пористости	0,52				#ДЕЛ/0!	#ДЕЛ/0!							
16	Плотность природ., г/см <sup>3</sup>	1,83									1,82			
17	Плотность сухого, г/см <sup>3</sup>	1,74												
18	Коэфф. водонасыщ	0,25									#ДЕЛ/0!			

Рис. 15. Окончательный вид примера 3 на листе Excel

Изменяя значения в ячейках **B1**, **B2** и **B3**, вы будете получать значения, как по результатам интерполяции таблицы, так и по результатам расчета. Как вы можете заметить, они немного отличаются. Расчетные более точные, в том числе и потому, что указывается реальная плотность частиц, полученная в результате лабораторных исследований.

**Примечание.** Вы могли заметить, что таблица заполнена не полностью. И даже если указать в ячейке **B1**, например 0,25, а в ячейке **B2** – 40, интерполяция все равно будет выполнена. Только это, конечно уже будет экстраполяция. Чтобы этого избежать, можно не использовать условные операторы, как в предыдущих примерах, а заполнить пустые ячейки, например, словом «Ошибка» (рис. 16).

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	Влажность природ. W, д.е.	0,250			p, г/см3, при сопротивлении qc, МПа									
2	qc (по статике), МПа	40		w	50	40	30	20	15	10	5	3	2	1
3	Плотность частиц, г/см3	2,65		0	1,83	1,8	1,78	1,73	1,72	1,68	1,61	1,57	1,55	1,5
4				0,05	1,92	1,89	1,87	1,82	1,8	1,76	1,7	1,65	1,63	1,58
5	Результат интерполяции			0,1	2	1,98	1,95	1,91	1,88	1,84	1,78	1,74	1,71	1,65
6				0,15	2,1	2,07	2,01	1,99	1,97	1,93	1,86	1,82	1,78	1,72
7	Плотность природ., г/см3	Ошибка		0,2	Ошибка	Ошибка	Ошибка	2,08	2,05	2,01	1,94	1,89	1,86	1,8
8	Плотность сухого, г/см3	#ЗНАЧ!		0,25	Ошибка	Ошибка	Ошибка	Ошибка	Ошибка	Ошибка	1,94	1,96	1,94	1,88
9	Коэфф. пористости	#ЗНАЧ!		0,3	Ошибка	Ошибка	Ошибка	Ошибка	Ошибка	Ошибка	1,96	1,96	1,95	1,95
10	Коэфф. водонасыщ.	#ЗНАЧ!												
11														
12							40	40			#ДЕЛ/0!			
13	Расчет плотности по Мариупольскому				0,25	Ошибка	Ошибка							
14					0,25	Ошибка	Ошибка							
15	Коэфф. пористости	0,47				#ДЕЛ/0!	#ДЕЛ/0!							
16	Плотность природ., г/см3	2,26									Ошибка			
17	Плотность сухого, г/см3	1,80												
18	Коэфф. водонасыщ	1,41									#ДЕЛ/0!			

Рис. 16. Вариант запрета экстраполяции, без использования условных операторов

В результате, в ячейке **B7**, вы получите «Ошибка» или #Н/Д.

**Выводы.** В статье были предложены некоторые варианты для поиска и интерполяции значений. Конечно, их гораздо больше. Но уже с помощью этой информации вы самостоятельно сможете интерполировать любые таблицы или значения, которые найдете в нормативной документации или полученные в результате исследований. Без использования специального программного обеспечения, созданного для целей обработки геологической информации, можно вполне обойтись MS Office Excel.

### Список литературы

1. Л.Г. Мариупольский. Исследования грунтов для проектирования и строительства свайных фундаментов. Москва, Стройиздат, 1989 г.
2. СП 22.13330.2016 «Основания зданий и сооружений» Актуализированная редакция СНиП 2.02.01-83\* (с изменениями №1, 2, 3). 2019 г.